# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: zoulnix <z[o]ulnix.borkedduck.c[o]m>

pkgname=libfakekey
pkgver=0.1
pkgrel=5
pkgdesc='X virtual keyboard library.'
arch=('i686' 'x86_64')
url='http://matchbox-project.org/'
license=('GPL')
depends=('libxtst')
options=('!libtool')
source=("http://matchbox-project.org/sources/${pkgname}/${pkgver}/${pkgname}-${pkgver}.tar.bz2")
md5sums=('83dbde4d77e8baf0176fe4291d8a2303')

build() {
  cd "${pkgname}-${pkgver}"

  ./configure --prefix=/usr --sysconfdir=/etc --localstatedir=/var --disable-static
  make AM_LDFLAGS=-lX11
}

package() {
  cd "${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
